

# SPARQL



Eirik Tvedt Jensen  
Kasper Rynning-Tønnesen

# SPARQL

- Hva er SPARQL?
- Hvorfor SPARQL?
- Triple Pattern
- Graph Pattern
- SPARQL 1.0
- SPARQL 1.1

# Hva er SPARQL

- Spørrespråk for RDF
- RDF-grafer
- IRI - ikke URI
- Semantisk spørrespråk
- Minner om SQL
- SPARQL Endpoint

# Hvorfor SPARQL?

- Store mengder RDF data
- Spørringer gir spesifikk informasjon
- Spørringer mot eksterne servere
- Applikasjoner kan bruke SPARQL spørringer

# Triple pattern

- SPARQL bygd på 3-tuple
- Forskjell RDF og SPARQL triples
- Variabler - ?variableName, \$variableName

# Graph Pattern

- Som triple pattern, mer kompleks
- Kolleksjon av triple pattern
- Må matche alt

```
{  
  ?who foaf:name ?name .  
  ?who foaf:interest ?interest .  
  ?who foaf:knows ?others .  
}
```

# SPARQL 1.0

- SPARQL 1.1 bygger på SPARQL 1.0
- Friend Of a Friend (FOAF)
- SELECT, ASK, DESCRIBE, CONSTRUCT

# SELECT

```
1: base <http://danbri.org/foaf.rdf>
2: prefix foaf: <http://xmlns.com/foaf/0.1/>

3: select *
4: from <http://danbri.org/foaf.rdf>
5: where
6: {
7:   <#danbri> ?property ?value.
8: }
```



# Avansert SELECT

```
base <http://danbri.org/foaf.rdf>
prefix foaf: <http://xmlns.com/foaf/0.1/>

select *
from <http://danbri.org/foaf.rdf>
where
{
    <#danbri> foaf:knows ?friend .
    optional { ?friend foaf:name ?name . }
    optional { ?friend foaf:mbox ?email . }
    optional { ?friend foaf:homepage ?homepage . }
}
```

# Avansert SELECT

```
base <http://danbri.org/foaf.rdf>
prefix foaf: <http://xmlns.com/foaf/0.1/>

select ?name ?email
from <http://danbri.org/foaf.rdf>
where
{
    ?x a foaf:Person.
    ?x foaf:name ?name.
    ?x foaf:mbox ?email.
}
order by asc(?name)
limit 10 offset 1
```

# Avansert SELECT

```
base <http://danbri.org/foaf.rdf>
prefix foaf: <http://xmlns.com/foaf/0.1/>

select ?name ?mbox
from <http://danbri.org/foaf.rdf>
where
{
    ?person a foaf:Person .
    ?person foaf:name ?name .
    {
        { ?person foaf:mbox ?mbox. }
        union
        { ?person foaf:mbox_shalsum ?mbox. }
    }
}
```

# Avansert SELECT

```
base <http://danbri.org/foaf.rdf>
prefix foaf: <http://xmlns.com/foaf/0.1/>
prefix xsd: <http://www.w3.org/2001/XMLSchema#>

select ?name ?dob
from <http://danbri.org/foaf.rdf>
where
{
    ?person a foaf:Person .
    ?person foaf:name ?name .
    ?person foaf:dateOfBirth ?dob .
    filter ( xsd:date(str(?dob)) >= "1970-01-01"^^xsd:date &&
           xsd:date(str(?dob)) < "1980-01-01"^^xsd:date )
}
```

# Avansert SELECT

```
1: prefix foaf: <http://xmlns.com/foaf/0.1/>
2: select distinct ?name ?email
3: from named <http://liyangyu.com/foaf.rdf>
4: from named <http://danbri.org/foaf.rdf>
5: where
6: {
7:     graph <http://liyangyu.com/foaf.rdf>
8:     {
9:         ?person a foaf:Person .
10:        ?person foaf:mbox ?email .
11:        optional { ?person foaf:name ?name . }
12:    } .
13:    graph <http://danbri.org/foaf.rdf>
14:    {
15:        ?person1 a foaf:Person .
16:        ?person1 foaf:mbox ?email .
17:        optional { ?person1 foaf:name ?name . }
18:    } .
19: }
```

# CONSTRUCT

```
prefix foaf: <http://xmlns.com/foaf/0.1/>

construct {
    ?person a foaf:Person.
    ?person foaf:name ?name.
    ?person foaf:mbox ?email.
}

from <http://danbri.org/foaf.rdf>

where
{
    ?person a foaf:Person.
    ?person foaf:name ?name.
    ?person foaf:mbox ?email.
}
```

# DESCRIBE

```
prefix foaf: <http://xmlns.com/foaf/0.1/>

describe ?x
from <http://danbri.org/foaf.rdf>
where
{
    ?x foaf:mbox <mailto:timbl@w3.org> .
}
```

# ASK

```
prefix foaf: <http://xmlns.com/foaf/0.1/>

ask
from <http://danbri.org/foaf.rdf>
where
{
    ?x foaf:mbox <mailto:danbri@danbri.org>.
}
```



# SPARQL 1.1

- Aggregate
  - COUNT
  - SUM
  - MIN/MAX
  - AVG
  - GROUP\_CONCAT
  - SAMPLE
- Negation
- Update
- Flytting

```
PREFIX : <http://books.example/>
SELECT ?org (SUM(?price) AS ?totalPrice)
WHERE {
    ?org :affiliates ?auth .
    ?auth :writesBook ?book .
    ?book :price ?price .
}
GROUP BY ?org
```

```
prefix foaf:<http://xmlns.com/foaf/0.1/>
prefix liyang: <http://www.liyangyu.com/foaf.rdf#>
INSERT DATA
{
    GRAPH <http://www.liyangyu.com/foaf.rdf>
    {
        liyang:liyang foaf:workplaceHomepage <http://www.delta.com> ;
        foaf:schoolHomepage <http://www.osu.edu> .
    }
}
```